
Nice to know

ANSYS

Im Customer Bereich unserer Homepage haben Sie vollen Zugriff auf unsere Supportdatenbank, in 
der alle uns bekannten Probleme bei der Anwendung von ANSYS und DESIGNSPACE dokumentiert 
sind.

Eine Übersicht zu den *get Kurzbefehlen findet man im APDL Programmers Guide> Appendix B>
*get Function Summary

Eine exakte Übersicht zu den Massen und den Massenträgheitsmomenten erhält man ohne die 
Berechnung durchlaufen zu müssen mit den psolve Komando. Dazu ist die folgende Befehlsfolge 
einzugeben:

irlf,-1
psolve,elform
psolve,elprep
irlist

Sehr praktisch für die Überprüfung von Kontaktpaaren ist, dass psolve,elform auch einen detaillierten 
Überblick über die verwendeten Parameter der Kontaktpaare ausgibt. Die Ausgabe ist sehr 
umfangreich und sollte deshalb in eine Datei mit /output umgeleitet werden. 

Unter Unix müssen alle Macronamen gross geschrieben werden (MYMAC.MAC).

Ein Reset der graphischen Darstellung wird mit /reset ausgeführt. Ein Reset der Datenbasis kann mit 
cdwrite und anschliessendem cdread ausgeführt werden.

In ANSYS 6.1 sollten beim Start die Default-Einstellungen für den Speicher verwendet werden. Damit 
ist in der Regel ein sicheres Durchlaufen des Jobs gewährleistet. Auch Zugriffsprobleme auf die 
integrierte Hilfe lassen sich damit vermeiden.

Der 3D-Treiber updated die Darstellung nicht automatisch. Weisse Flecken in den Abbildungen im 
Report Generator können behoben werden, indem beim Start von ANSYS nicht der 3D-Treiber 
ausgewählt wird.

Auf unserer Homepage finden Sie ein Inputfile zur Darstellung von Ergebnissen mit topographisch 
verzerrten Gitternetzlinien.

DesignSpace

In manchen Fällen kann es vorkommen, das unter UG 17/18 der Button zum Starten von
DesignSpace nicht vorhanden ist obwohl die Installation –scheinbar- erfolgreich verlaufen ist. 

Dieser Eintrag wird durch eine *.dll geregelt die durch die custom_dirs.dat angesprochen wird. Bei 
manchen Firmeninstallationen wird diese „lokale“ custom_dirs.dat durch eine „globale“ 
custom_dirs.dat überschrieben. 
Zum Beheben des Problems muss man in diese globale custom_dirs.dat folgenden Pfad manuell 
eintragen: XX:\Programme\ANSYS INC\CAD Integration\Unigraphics 

Hyperelastizität mit Mooney-Rivlin 
Materialgesetz

Problem:
Die Verwendung des Mooney-Rivlin Materialgesetzes für hyperelastisches Material kann wahlweise 
mit dem Befehl TB,MOONEY,,,, oder mit TB,HYPER,,,MOONEY implementiert werden.
Es stellt sich nun die Frage welche Materialparameter jeweils gefordert werden und welche 
Elementauswahl zu treffen ist.

Erläuterung:
Mit einem Mooney-Rivlin Gesetz kann hyperelastisches Materialverhalten gummiartiger Werkstoffe 
abgebildet werden, wenn Dehnungen zwischen 30% im Druckbereich und 100% im Zugbereich 
vorliegen. 
Für Gummimaterialien ist ferner typisch, dass sich im Laufe der Deformation das Volumen des 
Materials nicht ändert. Das Material verhält sich inkompressibel.
Quasi inkompressibles hyperelastisches Materialverhalten kann mit den älteren HYPERxx-
Elementen (56, 58, 74, 158) abgebildet werden. Für die Querkontraktionszahl
muss dann gelten �<0.5, also zum Beispiel �=0.499. 
Den exakt inkompressiblen Fall �=0.5  kann man mit diese Elementen nicht abbilden, wohl aber 
mit den neueren 18x-Elementen (182, 183, 185, 186, 187). Zur Abbildung exakter Inkompressibilität
ist hier jedoch nicht die Querkontraktion �=0.5 zu verwenden sondern der Parameter d=0. 
Allein von der Elementwahl hängt es dann ab, welches TB-Kommando zu verwenden ist und in 
welcher Form die Materialparameter vorzugeben sind.

Beispiel:
Ein Gummiblock wird um 100% in X-Richtung gedehnt. Ein Mooney-Rivlin-Gesetz mit zwei 
bekannten Parametern c10 und c01 soll verwendet werden. Die Modellierung mit HYPER58 und 
SOLID185 liefern praktisch identische Ergebnisse.
Es sei an dieser Stelle noch einmal gesagt, dass die Parameter des Mooney-Rivlin Gesetzes allein 
aus Versuchen bestimmt werden müssen und nicht etwa einem Tabellenwerk entnommen werden 
können.

Ein Teil des Eingabeskriptes für eine Modellierung mit HYPER58 und SOLID185 sei im Folgenden 
kurz skizziert: 
Außer den Mooney-Parametern muss nur noch die Inkompressibiliät beschrieben werden, etwa mit 
der Querkontraktionszahl �=0.499 (quasi inkompressibel) oder mit dem Parameter d=0.0 (exakt 
inkompressibel).
Ferner ist darauf zu achten, dass bei SOLID185 die U/P-Element-Formulierung mittels einer 
Keyoption aktiviert werden sollte, sobald man quasi oder exakt inkompressibles Materialverhalten 
modellieren möchte. Damit wird die numerische Stabilität gewährleistet.

Hyperelastizität mit Mooney-Rivlin 
Materialgesetz

ANSYS Eingabesatz (ANSYS 6.1):

!Hyperelastizität mit Mooney Rivlin
!Gesetz; Verwendung der SOLID185

fini
/clear

c10 = 0.293
c01 = 0.177
d   = 0.0

/prep7
block,0,10,0,10,0,10

et,1,solid185
keyopt,1,6,1

tb,hyper,1,1,2,mooney
tbdata,,c10,c01,d

esize,1.0
vmesh,all

/solu
nsel,s,loc,x,0
d,all,all,0
nsel,s,loc,x,10
d,all,ux,5
d,all,uy,0
d,all,uz,0

nlgeom,on
autots,off
nsub,10
allsel
solve

/post1
/dscale,1,1
plesol,s,x

!Hyperelastizität mit Mooney Rivlin
!Gesetz; Verwendung der HYPER58

fini
/clear

c10 = 0.293
c01 = 0.177
nu  = 0.499

/prep7
block,0,10,0,10,0,10

et,1,hyper58
mp,prxy,1,nu
tb,mooney,1,1
tbdata,,c10,c01

esize,1.0
vmesh,all

/solu
nsel,s,loc,x,0
d,all,all,0
nsel,s,loc,x,10
d,all,ux,10
d,all,uy,0
d,all,uz,0

nlgeom,on
autots,off
nsub,10
allsel
solve

/post1
/dscale,1,1
plesol,s,x

Einfache Lastdefinition mit Tabellen 
und Funktionen

Problem:
Während bei der Lastdefinition auf Flächen mit SFGRAD eine linear veränderliche Last angegeben werden 
kann, wird diese Funktionalität für Balken nicht angeboten. Das SFBEAM-Kommando bietet zwar die 
Möglichkeit eine veränderliche Last vorzugeben, aber nur beschränkt auf ein Balkenelement.
Hier empfiehlt es sich nun eine table-load zu verwenden, die in Abhängigkeit vom Ort den Belastungswert 
vorgibt.

Erläuterung:
Mit der Verwendung des Function-Editors ist es möglich auf sehr einfache Art und Weise eine Belastung oder 
Randbedingung, die einer bestimmten Funktion folgt, zu definieren. Für einfache Fälle kann die Definition des 
Tables auch per Hand erfolgen.
Den alten Balkenelemente (z.B. beam4) liegt die Bernoulli Theorie zugrunde. Für die Formfunktionen werden 
die Hermiteschen Polynome verwendet. Deshalb ergibt sich für Biegebelastung eine kubische Antwortform, 
die der analytischen Lösung entspricht. Der User ist also in der Lage die Struktur mit äußerst wenigen 
Elementen abzubilden.  
Bei den neuen Elementen wurde die Thimoshenko Theorie zugrundegelegt, die die Schubverzerrungen 
berücksichtigt. Es ergeben sich für das 188-Balkenelement lineare und für das 189-Balkenelement 
quadratische Antwortformen unter Biegung. Eine feine Diskretisierung ist notwendig.
In diesem Fall können veränderliche Lasten nur über die Table-Load definiert werden.
Für Belastungen, die einer komplizierteren Funktion folgen, empfiehlt es sich auf alle Fälle den Function-Editor 
zu verwenden, der diese Table-Load automatisch kreiert.

Beispiel:
Das Beispiel zeigt die einfache Verwendung der Table-Load Funktion zur  Definition einer linear 
veränderlichen Last auf einen spiralförmigen Balkenzug.
Es ist zu beachten, dass im *dim Befehl am Ende die Variablen definiert werden, von denen das Table 
abhängt. Es gibt folgende fest belegte Variablennamen: 

x,y,z Ortsangabe in globalen kartesischen Koordinaten
temp Temperatur
velocity Geschwindigkeit bei Fluid116
pressure Druck

Der Name des definierten Tables kann im übrigen auch wieder in einem anderen Table als Bezugsvariable 
eingesetzt werden. Auf diese Weise können Abhängigkeiten von mehreren Variablen formuliert werden. 
Z.B. Table1: Radius abhängig von x und y, Table2: Druck abhängig vom Radius

Einfache Lastdefinition mit Tabellen 
und Funktionen

ANSYS Eingabesatz (ANSYS 6.1):

fini
/clear

/prep7

et,1,188

sectype,10,beam,rect,
secdata,.1,.2

mp,ex,1,2e5
mp,prxy,.3
csys,1

*do,i,1,10,1
k,i,3,i*90,i
k,100+i,0,i*90,i
*enddo

*do,i,1,9,1
l,i,i+1
lsel,s,,,i
latt,1,1,1,,100+i,100+i+1,10
*enddo

/eshape,1
esize,.3
alls
lmesh,all
eplo

/solu
csys,0
ksel,s,,,1
nslk,s
d,all,all,0

eplo

csys,0
*dim,tpre,table,2,1,1,z
tpre(1,0)=0,10
tpre(1,1)=0,-30

alls
sfbeam,all,2,pres,%tpre%

solve

/post1
/eshape,0

/psf,pres,norm,2,0,1
eplo

Flußbestimmung bei Kanten-
Elementen (SOLID117)

Problem:
Neben der Bestimmung der Induktivitäten von Leiteranordnungen ist meist auch die Berechnung 
von magnetischen Flüssen Ziel einer Magnetfeldanalyse (statisch - dynamisch).

Erläuterung:
ANSYS liefert mit dem Makro LMATRIX eine Methode zur Berechnung der Induktivitäten, die auch 
den totalen Fluß mit ausgibt. Darüber hinaus können Flußverkettungen ebener Modellaufgaben mit 
dem Makro FLUXV an vorher bereitgestellten Pfaden bestimmt werden.
Für räumliche Modelle ist ein analoges Vorgehen nicht übertragbar, da hier über die Oberfläche 
integriert werden muss. Sofern die Magnetfeldberechnung aber mit dem Element SOLID117 (Edge-
Element) erfolgt, zeigt das beiliegende Makro eine einfache und zudem sehr genaue 
Berechnungsmöglichkeit von Flüssen. 
Grundlage dieser Auswertung ist die Tatsache, dass der Freiheitsgrad AZ dieses Elementes direkt 
den Flußbeitrag der Elementkante abbildet. Somit müssen zur Auswertung lediglich die 
entsprechenden Knotenbeiträge summiert werden. Wichtig hierbei ist die korrekte Berücksichtigung 
der Richtung der Kante, die das Vorzeichen des Flußbeitrages bestimmt.

Beispiel:
Zur Verifikation des Makros liegt ein einfaches Modell eines Ringleiters vor, welcher von einem 
Einheitsstrom angeregt wird. Nach erfolgter Berechnung kann das Makro aufgerufen werden. Hier 
selektieren Sie bitte eine entsprechende Berandungslinie (Kreisbogen) des Leiters und definieren 
den Startpunkt. Im Ergebnis erhalten Sie den Fluß durch den aufgespannten Modellbereich, der 
dem halben Gesamtfluß entspricht, da es ein Halbmodell ist.

Flußbestimmung bei Kanten-
Elementen (SOLID117)

ANSYS Eingabesatz (ANSYS 6.1):

!/pmacro
! Voraussetzung: erfolgte Lösung mit solid Model
/uis,msgpop,2
lplo
*msg,warn
Picken Sie die Pfadlinie!
lsel,s,p
cm,pfad_l,line ! Linien f. Pfad
nsll,s,1
cm,pfad_n,node
*get,_nmax,node,,count
*msg,warn
Picken Sie den Startknoten des Pfades!
nsel,r,p
*get,_np1,node,,num,max ! Startknoten
cmsel,s,pfad_n

! Voreinstellungen
zahl=(_nmax-1)/2
_phi=0
_tt=
*dim,_tt,array,zahl,2

*do,jj,1,(_nmax-1)/2

_np2=nnear(_np1) ! nachfolgender Mittenknoten
nsel,u,node,,_np1 ! Startknoten weg

_np3=nnear(_np2) ! 2. Eckknoten

*if,_np3,gt,_np1,then ! Vorzeichencheck
_a=az(_np2)

*else
_a=-az(_np2)

*endif

_phi_p=_a ! edge flux

! debuginfo
_tt(jj,1)=_a ! Flussbeitrag
_tt(jj,2)=_np2 ! Knotennummer

! Flusssummation
_phi=_phi+_phi_p

! nächster Step
cmsel,s,pfad_n
nsel,u,node,,_np2
nsel,u,node,,_np1
cm,pfad_n,node
_np1=_np3

*enddo

*stat,_phi

Ein aufrufendes Inputfile finden Seite 
auf unserer Homepage

Kombinierte Strahlung und 
Wärmeleitung durch ein Medium

Problem:
Innerhalb eines geschlossenen Gehäuses soll der Wärmetransport durch Wärmeleitung und 
Strahlung berechnet werden.

Erläuterung:
Will man die Temperaturverteilung in einem geschlossenen mit Luft gefüllten Behälter untersuchen, 
so müssen neben der meist kleinen Wärmeleitung in der Luft auch Konvektion und Strahlung 
berücksichtigt werden. Während die Berechnung der Konvektion eine  Strömungsberechnung mit 
ANSYS/FLOTRAN erforderlich macht, lassen sich Wärmeleitung  und Strahlung mit den normalen 
ANSYS-Temperaturfeldelementen berechnen.
Es gibt im ANSYS zwei Methoden zur Berechnung der Strahlung, die Radiosity Methode und  die 
Matrixmethode. Nur letztere kann zur Modellierung der Strahlung innerhalb eines  
Berechnungsgebietes verwendet werden. Die Strahlung austauschenden Ränder werden durch 
link32 bzw. shell57 Elemente identifiziert. Mit Hilfe dieser Elemente wird das Superelement 
aufgestellt, welches dann den Strahlungsaustausch zwischen den betroffenen Knoten  liefert. Dabei 
ist es völlig unerheblich zwischen welchen Knoten der Strahlungsaustausch stattfindet, nach 
Aussen oder nach Innen.   

Beispiel:
Betrachtet wird ein einfacher viereckiger Raum, der von einer Wand umschlossen wird. Im Inneren 
ist die Wärmeleitung sehr viel kleiner als in der umgebenden Wand. Auf der linken Seite wird die
Temperatur auf 100 Grad gehalten, während auf der rechten Seite ein Waermestrom von 1000 
W/m² aufgeprägt wird. In den resultierenden Temperaturverteilungen erkennt man sehr deutlich den
grossen Einfluss der Strahlung. Ohne Strahlung ist die Temperatur sehr viel höher, weil der 
Wärmetransport fast ausschliesslich infolge von Wärmeleitung durch die Wand erfolgt.

ANSYS Eingabesatz (ANSYS 6.1):

!***************************************
!Strahlung und Wärmeleitung
!im Inneren eines Bauteils
!Strahlung mit Matrixmethode
!***************************************

fini
/clear
/filna,radiation

/prep7

rect,0,1,0,1
rect,-0.1,1.1,-0.1,1.1
aovl,all

asel,s,,,1
aatt,1
asel,s,,,3
aatt,2

alls

et,1,55

esiz,0.05
amesh,all

mp,kxx,1,0.01
mp,kxx,2,20

et,2,32
type,2
lsel,s,,,1,4,1
nsll,s,1
lmesh,all

nsel,s,loc,x,-0.1
d,all,temp,100

nsel,s,loc,x,1.1
!d,all,temp,400
sf,all,hflux,1000

alls

!Strahlung
/aux12
lsel,s,,,1,4,1
nsll,s,1
esln,s,1
emis,1,0.8
emis,2,0.8
stef,5.67e-8
geom,1
vtype,1,200
write,strahl,sub
alls

/prep7
et,3,50
keyopt,3,1,1
type,3
se,strahl,sub

tunif,200
toffs,273

esel,u,ename,,32

/solu
solve

/post1
set,last
plnsol,temp

Kombinierte Strahlung und 
Wärmeleitung durch ein Medium

Aktuelle Seminartermine:

Ermüdung unter mehrachsiger Beanspruchung
03.09.02 in Grafing bei München

Palmyra
04.-05.09.02 in Grafing bei München

ANSYS Kontaktberechnungen 
04.-06.09.02 in Leinfelden-Echterdingen bei Stuttgart

Schnuppertraining: Konstruktionsbegleitend berechnen – DesignSpace
10.09.02 in Leinfelden-Echterdingen bei Stuttgart

FE-Idealisierung in der Strukturmechanik
10.-13.09.02 in Grafing bei München

Einführung in FALANCS
17.-19.09.02 in Grafing bei München

Betriebsfestigkeit mit ANSYS und FALANCS
20.09.02 in Grafing bei München

Einführung in ANSYS – Strukturmechanik
17.-20.09.02 in Leinfelden-Echterdingen bei Stuttgart

Effizientes Arbeiten mit ANSYS
24.-27.09.02 in Burgdorf bei Hannover

Plastizität und Kriechen
26.-27.09.02 in Leinfelden-Echterdingen bei Stuttgart

Weitere Veranstaltungen:

VDI-Tagung: Berechnung und Simulation im Fahrzeugbau
01.-02.09.02 in Würzburg

11. Aachener Kolloquium Fahrzeug- und Motorentechnik
07.-09..09.02 in Aachen

An event to remember:

20. CADFEM Users` Meeting 2002
International Congress on FEM Technology
October 9-11, 2002
Kultur- und Congress-Centrum "Graf-Zeppelin-Haus"
Friedrichshafen, Germany

Termine rund um CADFEM


